

NAG Library Routine Document

F08GEF (DSPTRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08GEF (DSPTRD) reduces a real symmetric matrix to tridiagonal form, using packed storage.

2 Specification

```
SUBROUTINE F08GEF(UPLO, N, AP, D, E, TAU, INFO)
INTEGER          N, INFO
double precision AP(*), D(N), E(N-1), TAU(N-1)
CHARACTER*1      UPLO
```

The routine may be called by its LAPACK name *dsptrd*.

3 Description

F08GEF (DSPTRD) reduces a real symmetric matrix A , held in packed storage, to symmetric tridiagonal form T by an orthogonal similarity transformation: $A = QTQ^T$.

The matrix Q is not formed explicitly but is represented as a product of $n - 1$ elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 8).

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: UPLO – CHARACTER*1 *Input*
On entry: indicates whether the upper or lower triangular part of A is stored.
UPLO = 'U'
The upper triangular part of A is stored.
UPLO = 'L'
The lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 3: AP(*) – **double precision** array *Input/Output*
Note: the dimension of the array AP must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the n by n symmetric matrix A , packed by columns.

More precisely,

if UPLO = 'U', the upper triangle of A must be stored with element A_{ij} in $AP(i + j(j - 1)/2)$ for $i \leq j$;

if UPLO = 'L', the lower triangle of A must be stored with element A_{ij} in $AP(i + (2n - j)(j - 1)/2)$ for $i \geq j$.

On exit: AP is overwritten by the tridiagonal matrix T and details of the orthogonal matrix Q .

4: D(N) – **double precision** array *Output*

On exit: the diagonal elements of the tridiagonal matrix T .

5: E(N – 1) – **double precision** array *Output*

On exit: the off-diagonal elements of the tridiagonal matrix T .

6: TAU(N – 1) – **double precision** array *Output*

On exit: further details of the orthogonal matrix Q .

7: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $(A + E)$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n , and ϵ is the **machine precision**.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^3$.

To form the orthogonal matrix Q F08GEF (DSPTRD) may be followed by a call to F08GFF (DOPGTR):

```
CALL DSPTRD(UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
```

To apply Q to an n by p real matrix C F08GEF (DSPTRD) may be followed by a call to F08GGF (DOPMTR). For example,

```
CALL DOPMTR('Left',UPLO,'No Transpose',N,P,AP,TAU,C,LDC,WORK,
+          INFO)
```

forms the matrix product QC .

The complex analogue of this routine is F08GSF (ZHPTRD).

9 Example

This example reduces the matrix A to tridiagonal form, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage.

9.1 Program Text

```
*      F08GEF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*      .. Local Scalars ..
      INTEGER          I, INFO, J, N
      CHARACTER        UPLO
*      .. Local Arrays ..
      DOUBLE PRECISION AP(NMAX*(NMAX+1)/2), D(NMAX), E(NMAX-1),
+      TAU(NMAX-1)
*      .. External Subroutines ..
      EXTERNAL        DSPTRD
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08GEF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read A from data file
*
          READ (NIN,*) UPLO
          IF (UPLO.EQ.'U') THEN
              READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
          ELSE IF (UPLO.EQ.'L') THEN
              READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
          END IF
*
*          Reduce A to tridiagonal form
*
          CALL DSPTRD(UPLO,N,AP,D,E,TAU,INFO)
*
*          Print tridiagonal form
*
          WRITE (NOUT,*)
          WRITE (NOUT,*) 'Diagonal'
          WRITE (NOUT,99999) (D(I),I=1,N)
          WRITE (NOUT,*) 'Off-diagonal'
          WRITE (NOUT,99999) (E(I),I=1,N-1)
      END IF
*
99999  FORMAT (1X,8F9.4)
      END
```

9.2 Program Data

```
F08GEF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  2.07
  3.87 -0.21
  4.20 1.87 1.15
 -1.15 0.63 2.06 -1.81          :End of matrix A
```

9.3 Program Results

F08GEF Example Program Results

Diagonal

2.0700	1.4741	-0.6492	-1.6949
--------	--------	---------	---------

Off-diagonal

-5.8258	2.6240	0.9163
---------	--------	--------
